

# Hướng dẫn: Power Apps gọi Power Automate để tìm kiếm SharePoint List (có phân trang)

Tài liệu mẫu để copy/paste Expressions + Power Fx • Cập nhật 06/01/2026

---

## 1. Tổng quan

Mục tiêu: từ Canvas App (Power Apps) gọi một Flow (Power Automate) để tìm kiếm dữ liệu trong SharePoint List theo các tham số root, lv1, lv2, searchText, pageSize, pageIndex và một mảng lv1Array. Flow sẽ thực hiện: Get items (lọc OData) -> Filter array (lọc nâng cao) -> phân trang -> trả JSON về Power Apps.

Khuyến nghị: lọc được gì bằng OData trong Get items thì làm trước (nhanh hơn). Những điều kiện khó như "contains" nhiều cột hoặc "giá trị nằm trong mảng" thì làm bằng Filter array sau.

## 2. Chuẩn bị SharePoint List

Ví dụ List tên Requests với các cột (internal name): Root, Lv1, Lv2, Title (mặc định), và (tùy chọn) Description.

Lưu ý quan trọng: trong OData/Flow bạn phải dùng internal name của cột. Nếu cột hiển thị có dấu cách (ví dụ "Level 1"), internal name có thể là Level\_x0020\_1. Bạn kiểm tra internal name bằng cách mở cột trong List settings và xem tham số Field= trên URL.

## 3. Tạo Flow trong Power Automate

### 3.1 Trigger và inputs (Power Apps V2)

Tạo Instant cloud flow với trigger Power Apps (V2) và thêm 7 inputs:

- root (Text)
- lv1 (Text)
- lv2 (Text)
- searchText (Text)
- pageSize (Number)
- pageIndex (Number)
- lv1ArrayJson (Text) - JSON string từ Power Apps

### 3.2 Initialize variables + Parse JSON

Tạo 7 biến: 6 biến scalar (string/int) và 1 biến array. lv1Array nên gửi từ Power Apps dưới dạng JSON string để Parse JSON ra mảng.

```
// Initialize variable (recommended: use coalesce to avoid null)
varRoot      (String) = coalesce(triggerBody()?['root'], '')
varLv1       (String) = coalesce(triggerBody()?['lv1'], '')
varLv2       (String) = coalesce(triggerBody()?['lv2'], '')
varSearchText(String) = coalesce(triggerBody()?['searchText'], '')
varPageSize  (Integer) = int(coalesce(triggerBody()?['pageSize'], 20))
varPageIndex (Integer) = int(coalesce(triggerBody()?['pageIndex'], 0))
varLv1Array  (Array)  = [] // set after Parse JSON
```

Parse JSON cho input lv1ArrayJson. Nếu Power Apps gửi dạng collection object như [{ "Value": "A" }, { "Value": "B" }] thì dùng schema sau:

```
{
  "type": "array",
  "items": {
    "type": "object",
    "properties": {
      "Value": { "type": "string" }
    },
    "required": ["Value"]
  }
}
```

Sau Parse JSON, dùng Select để lấy mảng string và set vào varLv1Array:

```
// Select (Data Operations)
// From: body('Parse_JSON')
// Map: Value -> item()?['Value']

// Set variable: varLv1Array = outputs('Select') // result: ["A","B"]
```

### 3.3 Build OData Filter Query cho Get items

Dùng Compose để tạo chuỗi OData. Mẫu dưới đây lọc Root/Lv1/Lv2 nếu có giá trị; nếu rỗng thì bỏ qua. Cuối chuỗi thêm điều kiện an toàn ID gt 0 để tránh dư "and".

```
concat(
  if(equals(variables('varRoot'), ''), '',
    concat("Root eq '", replace(variables('varRoot'),' ',''), "' and ")
  ),
  if(equals(variables('varLv1'), ''), '',
    concat("Lv1 eq '", replace(variables('varLv1'),' ',''), "' and ")
  ),
  if(equals(variables('varLv2'), ''), '',
    concat("Lv2 eq '", replace(variables('varLv2'),' ',''), "' and ")
  ),
  "ID gt 0"
)
```

### 3.4 Get items (SharePoint)

Thêm action Get items và điền:

- Site Address: site của bạn
- List Name: Requests (hoặc list của bạn)
- Filter Query: outputs('ODataFilter') (Compose ở bước 3.3)
- Top Count: tùy nhu cầu; bật Pagination nếu cần

### 3.5 Filter array (searchText + lv1Array)

Dùng Filter array để lọc nâng cao (ví dụ searchText contains nhiều cột + kiểm tra Lv1 nằm trong mảng). Trong Filter array: From = body('Get\_items')?['value']. Chuyển sang Advanced mode và dán expression sau:

```
@and(
  /* searchText: empty -> pass; not empty -> contains in Title + Description */
  or(
    equals(variables('varSearchText'), ''),
    contains(
      toLower(concat(string(item()?['Title']), ' ', string(item()?['Description']))),
      toLower(variables('varSearchText'))
    )
  ),
  /* lv1Array: empty array -> pass; otherwise Lv1 must be in the array */
  or(
    equals(length(variables('varLv1Array')), 0),
    contains(variables('varLv1Array'), string(item()?['Lv1']))
  )
)
```

Nếu bạn không có cột Description, hãy bỏ phần concat với Description.

### 3.6 Phân trang (pageSize/pageIndex)

Tính tổng và cắt trang từ kết quả Filter array:

```
// TotalCount
length(body('Filter_array'))

// PagedItems (pageIndex starts at 0)
take(
  skip(body('Filter_array'), mul(variables('varPageIndex'), variables('varPageSize'))),
  variables('varPageSize')
)
```

### 3.7 Respond JSON về Power Apps

Cách đơn giản: trả về một chuỗi JSON (Text) gồm total, pageIndex, pageSize, items.

```
// Compose: ResponseJson
json(
  concat(
    '{',
    '"total":', string(outputs('TotalCount')), ',',
    '"pageIndex":', string(variables('varPageIndex')), ',',
    '"pageSize":', string(variables('varPageSize')), ',',
    '"items":', string(outputs('PagedItems')),
    '}'
  )
)

// Respond to a PowerApp or flow
// Output: result (Text) = string(outputs('ResponseJson'))
```

## 4. Power Apps: Add Flow và gọi Run()

### 4.1 Add Flow vào Canvas App

Trong Power Apps Studio: mở app -> tab Action -> Power Automate -> chọn Flow bạn vừa tạo để add/attach.

### 4.2 Gọi Flow với 7 tham số (Power Fx)

Giả sử bạn có collection colLv1 dạng: { Value: "A" }, { Value: "B" }. Gọi Flow và gửi mảng bằng JSON(colLv1, JSONFormat.Compact).

```
// Button.OnSelect (ví dụ)
Set(
    varRes,
    SearchSPRequests.Run(
        ddRoot.Selected.Value,
        ddLv1.Selected.Value,
        ddLv2.Selected.Value,
        txtSearch.Text,
        Value(ddPageSize.Selected.Value),
        varPageIndex,
        JSON(colLv1, JSONFormat.Compact)
    )
);
```

### 4.3 Parse JSON trả về và bind vào Gallery

```
Set(varObj, ParseJSON(varRes.result));
Set(varTotal, Value(varObj.total));

ClearCollect(
    colData,
    ForAll(
        varObj.items,
        {
            ID: Value(ThisRecord.Value.ID),
            Title: Text(ThisRecord.Value.Title),
            Root: Text(ThisRecord.Value.Root),
            Lv1: Text(ThisRecord.Value.Lv1),
            Lv2: Text(ThisRecord.Value.Lv2)
        }
    )
);

// Gallery.Items = colData
```

## 5. Checklist lỗi thường gặp

- Filter Query không chạy: dùng sai internal name của cột hoặc chưa escape dấu nháy đơn (').
- Parse JSON lỗi: lv1ArrayJson không đúng schema (mảng string vs mảng object). Kiểm tra JSON thực tế trong Power Apps.
- Chậm/timeout: list lớn mà bạn kéo quá nhiều item về rồi mới Filter array. Ưu tiên lọc Root/Lv1/Lv2 bằng OData trước.

- Phân trang sai: thống nhất pageIndex bắt đầu từ 0 hay 1, rồi chỉnh công thức skip tương ứng.

Gợi ý thực hành: chạy Flow một lần trong Power Automate (Test) để xem output JSON, sau đó mới tích hợp vào Power Apps để giảm thời gian debug.